# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=tasty-hedgehog
pkgname=haskell-tasty-hedgehog
pkgver=1.3.1.0
pkgrel=51
pkgdesc="Integration for tasty and hedgehog"
url="https://github.com/qfpl/tasty-hedgehog"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-hedgehog' 'haskell-tagged' 'haskell-tasty')
makedepends=('ghc')
checkdepends=('haskell-tasty-expected-failure')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('2b8779487ce90f548576d82f1ebd694ee552ed7265d1426087b07e7cad3c1ea380ab0833b1d7a03bcbbcf4924226f166993fff9ae4d497cb10dbc4785a5c0807')

build() {
    cd $_hkgname-$pkgver

    if (( CHECKFUNC )); then
        _opts=('--enable-tests')
    else
        _opts=('--disable-tests')
    fi
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname "${_opts[@]}" \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 "LICENCE" "${pkgdir}/usr/share/licenses/${pkgname}/LICENCE"
    rm -f "${pkgdir}/usr/share/doc/${pkgname}/LICENCE"
}
